

# Maple à l'oral de Centrale

5 Juin 2012 - Version 0.2

MP\*

## A Présentation de l'épreuve

L'épreuve de Maths II est une épreuve de mathématiques assistée par ordinateur, en l'occurrence par l'usage d'un logiciel de calcul formel. Elle comporte un seul exercice qui est préparé pendant 30 minutes avec accès libre à l'ordinateur ; puis le candidat vient présenter pendant 30 minutes ses résultats et poursuivre la résolution de l'exercice au tableau. Une fois au tableau, il est souhaitable d'indiquer succinctement les questions qui ont été élucidées pendant la préparation et ensuite de ne pas trop perdre de temps sur les questions élémentaires, pour arriver au coeur du sujet. De façon générale, le candidat doit être très attentif aux conseils et a fortiori aux indications données oralement, et surtout de ne pas s'enfermer dans une tentative de résolution si l'examineur lui indique qu'elle risque de conduire à une impasse : on aboutit à un échec donc à une perte de temps qui empêchera la résolution d'autres questions. Il est essentiel que l'oral reste un dialogue, et une très bonne note peut être attribuée à un candidat qui, sans avoir résolu l'exercice lors de la préparation, aura montré une bonne réactivité aux indications proposées. Une des difficultés de cette épreuve est de savoir gérer l'aide que peut apporter le logiciel pour résoudre la question mathématique posée. Les exercices comportent pour la plupart au moins une question à résoudre avec l'outil informatique. Il est souvent attendu de pouvoir émettre une conjecture, qui sera démontrée dans la suite de l'exercice. Parfois, l'énoncé conseille d'utiliser le logiciel à bon escient au cours de l'exercice, sans qu'il soit imposé pour une question précise : il revient alors au candidat d'évaluer les questions où l'ordinateur lui apportera une aide précieuse. C'est par exemple le cas pour des calculs auxiliaires de développements limités ou d'intégrales élémentaires (coefficients de Fourier ...), ou des résolutions d'équations lors d'un exercice de géométrie : le logiciel permet d'alléger les calculs et d'éviter les erreurs liées au stress de l'épreuve. Il faut par contre savoir être circonspect sur certaines réponses du logiciel : que penser d'un candidat qui ne réagit pas face à l'affichage de valeurs propres « complexes » pour une matrice symétrique réelle, alors qu'une demande de valeurs approchées montre que leurs parties imaginaires sont des  $10^{-6}i$ ? Il ne s'agit pas là de savoir comment fonctionne le logiciel mais d'avoir un minimum de recul ou de prudence sur l'affichage de certains résultats.

[...]

## Commentaires et conseils concernant l'usage du logiciel de calcul formel

L'impression générale est ici un peu mitigée. Et il semble utile de redonner une liste de « savoir-faire » déjà publiée dans le rapport 2009 et qui figure en Annexe. Le jury a vu avec plaisir environ un quart des candidats très bien préparés, très à l'aise avec cet aspect de l'épreuve. Une majorité a montré l'habitude de côtoyer le logiciel et la connaissance des commandes usuelles. Signalons qu'il n'est d'ailleurs attendu ni dextérité ni connaissance savante des options possibles d'une fonction prédéfinie ; et il est normal et raisonnable qu'un candidat s'assure du bon emploi d'une fonction du logiciel en ayant recours à l'aide en ligne (encore faut-il en connaître le nom... : alors pourquoi se priver très souvent de l'affichage à l'écran des fonctions résidant dans une librairie, par un `with(...)` : au lieu d'un `with(...)` ; du logiciel Maple ?). Par contre, il faut constater et regretter qu'un trop grand nombre –même si c'est une minorité– continue à espérer découvrir les fonctions de base avec un usage fébrile de l'aide en ligne pendant la préparation : cela se traduit par une lourde perte de temps, et le bilan est en général catastrophique : aucun résultat ne sort d'une succession de lignes de code dont la plupart ont été rejetées par le logiciel... À la moindre sollicitation de l'examineur pour apporter les premières modifications permettant l'obtention d'au moins un résultat partiel (faute de temps, il ne peut être question de corriger les fautes accumulées),

la réponse « je n'ai pas pratiqué le logiciel cette année » ne peut constituer une excuse. L'usage d'un logiciel de calcul formel figure pour tous, au programme des deux années de classes préparatoires, et les candidats savent qu'ils en auront besoin lors de l'épreuve de maths II du concours Centrale. Le jury sanctionnera plus nettement cette attitude désinvolte : on ne peut pas arriver à cette épreuve si on ne sait pas utiliser efficacement le logiciel. Le peu de programmation attendue ne dépasse pas l'écriture de boucles avec d'éventuelles instructions conditionnelles ! Il ne s'agit pas d'un exercice d'algorithmique. Relevons quelques pratiques maladroites. On regrette d'abord un recours trop systématique à l'écriture de nombreuses et inutiles procédures, quelle que soit la complexité de 32 Épreuves orales Rapport du jury 2010 - Filière MP la question posée : cette démarche peut sembler tout à fait honorable, mais conduit trop souvent hélas à un échec, et donc à l'absence de résultats effectifs ; or c'est le but attendu. Et puis c'est une mauvaise compréhension de l'intérêt d'un logiciel de calcul formel : les fonctions prédéfinies sont là pour gagner du temps, et écrire une procédure pour définir la fonction « factorielle » (exemple caricatural mais vu cette année) n'a aucun intérêt ici. En fait, il est rarement indispensable d'écrire une procédure lors des questions proposées, ce qui ne signifie pas qu'elles sont parfois bienvenues ; mais l'écriture directe d'une boucle est souvent suffisante. Il est nécessaire de savoir distinguer la manipulation des « fonctions » et des « expressions », et d'estimer quand l'usage des unes ou des autres est plus favorable. Signalons que l'usage des expressions est souvent plus souple lorsqu'il doit se doubler de la création d'un opérateur mathématique qui manipule ces expressions. Certains ne savent d'ailleurs pas créer une suite, ou une fonction ; l'usage des crochets ou des parenthèses est mal compris :  $u[n]$  ou  $u(n)$  ? Et plus gênant encore est de voir écrire des tentatives du type  $u(n) := \dots$  ou  $f(x) := \dots$  pour fabriquer une fonction, et des candidats qui sont surpris que le logiciel ne réponde pas à leur attente !! Le logiciel met à disposition des outils commodes pour créer des séquences de résultats. Combien de fois on a vu le recours à des copier-coller quand l'énoncé demandait une séquence d'une dizaine ou d'une vingtaine de résultats (nécessaires à l'ébauche d'une conjecture « fiable ») ? Il faut enfin savoir indiquer au logiciel qu'une variable est par exemple entière, réelle positive, etc... Et connaître quelques commandes qui simplifient ou convertissent ou transforment des expressions sous une forme souhaitée.

Mais voici quelques points encore trop mal maîtrisés :

- \* la construction de matrices de taille variable. Il faut savoir fabriquer une fonction « définissante » des coefficients. On a ainsi vu régulièrement des candidats contraints d'écrire une matrice  $10 \times 10$  en tapant les 100 coefficients (dont beaucoup étaient nuls heureusement) !
- \* savoir obtenir des valeurs approchées des racines d'une équation, savoir que l'affichage d'un seul résultat numérique ne se traduit pas nécessairement par l'unicité d'une solution... ;
- \* pour le graphisme, il faut savoir superposer sur un même schéma divers types de graphes ;
- \* dans le cas particulier des équations différentielles, beaucoup ne savent pas visualiser le graphe d'une solution, lorsque le logiciel n'en donne pas une expression exacte.
- \* proscrire l'ouverture et l'usage simultanés des bibliothèques Maple `linalg` ET `LinearAlgebra` ;
- \* connaître les inconvénients ou avantages respectifs des commandes « `sum` » et « `add` » ...

[...]

## Liste de savoir-faire conseillés pour l'épreuve assistée par un logiciel de calcul formel

### Calcul algébrique (entiers, polynômes, équations)

- \* savoir calculer le quotient, le reste dans une division euclidienne dans  $\mathbf{Z}$ , dans  $\mathbf{Q}[X]$  ;
- \* savoir tester qu'un entier est premier, savoir travailler modulo  $n$  ;
- \* savoir factoriser (dans  $\mathbf{Q}[X]$  et éventuellement dans une extension simple suggérée par l'énoncé), développer, ordonner un polynôme ;

- \* savoir obtenir tous les coefficients, ou des coefficients précis d'un polynôme ;
- \* savoir calculer le pgcd de deux entiers, de deux polynômes ;
- \* savoir obtenir un couple donnant la relation de Bézout ;
- \* savoir déterminer les racines d'une équation (algébrique ou non) de façon exacte, de façon approchée ; savoir déterminer une valeur approchée d'une racine localisée dans un intervalle ;
- \* savoir décomposer une fraction rationnelle en éléments simples dans  $\mathbf{Q}(X)$  (éventuellement dans une extension simple de  $\mathbf{Q}$  suggérée par l'énoncé).

### Calcul matriciel

- \* savoir construire une matrice dont les coefficients sont donnés par une formule fonction du couple  $(i, j)$ , et dont la taille peut être variable (il ne peut être question de se limiter à savoir entrer une matrice  $3 \times 3$  par ses neuf coefficients) ;
- \* savoir calculer des produits matriciels, créer une matrice diagonale et a fortiori la matrice identité, former la transposée ;
- \* savoir calculer le rang, le noyau ou l'image (en obtenant une base de ces sous-espaces) ;
- \* savoir calculer le déterminant, éventuellement l'inverse, la comatrice (ou sa transposée) d'une matrice carrée ;
- \* savoir calculer le polynôme caractéristique d'une matrice carrée, ses valeurs propres, ses vecteurs propres ;
- \* savoir résoudre une équation d'inconnue matricielle (après l'avoir transformée en un ensemble d'équations scalaires d'inconnues les coefficients) ;
- \* savoir calculer le produit scalaire, le produit vectoriel de deux vecteurs de  $\mathbf{R}^3$ .

### Fonctions d'une ou plusieurs variables réelles, calcul différentiel, calcul intégral

- \* savoir composer des fonctions (ou des opérateurs), calculer des dérivées d'ordre supérieur à un ;
- \* savoir calculer un développement limité, savoir extraire la partie régulière d'un tel développement ;
- \* savoir calculer une intégrale de façon exacte, de façon approchée, faire un changement de variable ou une intégration par parties ;
- \* comprendre pourquoi le logiciel n'affiche pas toujours une limite explicite, ou le résultat d'un calcul d'intégrale, par manque d'information sur la nature d'un paramètre introduit : savoir préciser à quelle partie de  $\mathbf{R}$  il appartient (entier, réel positif... )

### Suites et séries numériques, suites et séries de fonctions

- \* savoir expliciter les premiers termes (de façon exacte ou approchée) d'une suite numérique ou d'une suite de fonctions, en particulier lorsqu'elle est définie par récurrence ;
- \* savoir obtenir un développement asymptotique d'une suite (fonction explicite de  $n$ ) ;
- \* savoir calculer les coefficients de Fourier d'une fonction périodique ;
- \* savoir visualiser sur un même schéma les premiers termes d'une suite de fonctions.

### Équations différentielles

- \* savoir résoudre une équation différentielle, un système d'équations différentielles, avec ou sans conditions initiales ;
- \* savoir récupérer une fonction solution et la tracer ;
- \* savoir tracer directement le graphe d'une solution obtenue par résolution numérique.

## Graphisme

On a déjà évoqué le tracé de graphes de fonctions d'une variable réelle, de solutions d'une équation différentielle.

- \* savoir tracer une courbe du plan, définie par une équation cartésienne (de façon implicite), ou par un paramétrage, peut-être en coordonnées polaires, et gérer les discontinuités ;
- \* savoir tracer une courbe paramétrée de l'espace ;
- \* savoir tracer une surface définie par un paramétrage, ou par une équation cartésienne ;
- \* savoir visualiser un ensemble de points, sous forme d'une ligne polygonale ou non.

## B Échauffement militaire : Maple à Saint-Cyr

**Exercice 1.** (Polynôme) On considère quatre réels  $m \neq n$  et  $x \neq y$ . Montrer qu'il existe un unique polynôme de degré 3 dont les extrema locaux valent  $m$  et  $n$  en  $x$  et  $y$  respectivement.

Plusieurs méthodes ici :

```
> P := x -> a*x^3 + b*x^2 + c*x + d;
> S := solve({P(x)=m, P(y)=n, D(P)(x)=0, D(P)(y)=0}, {a, b, c, d});
> factor(%1);
```

Ou bien :

```
> L1 := [CoefficientList(P(x), [a, b, c, d])];
> L2 := [CoefficientList(P(y), [a, b, c, d])];
> L3 := [CoefficientList(D(P)(x), [a, b, c, d])];
> L4 := [CoefficientList(D(P)(y), [a, b, c, d])];
> with(linalg):
> m := matrix([L1, L2, L3, L4]);
> factor(det(m));
```

Le polynôme est-il forcément de degré 3 ?

**Exercice 2.** (Matrice) On considère  $H = (h_{i,j})_{(i,j) \in \llbracket 1,3 \rrbracket^2}$  avec  $h_{i,j} = \frac{1}{|i-j|+1}$ . Montrer que l'application  $X \in \mathbb{R}^3 \mapsto \sqrt{t} X H X$  est une norme.

On définit facilement une matrice :

```
> M := n -> matrix(n, n, (i, j) -> (abs(i-j) + 1));
```

Le problème revient à montrer que les valeurs propres sont strictement positives :

```
> evalf(eigenvalues(M(3)));
```

**Exercice 3.** (Gram-Schmidt) Montrer que  $(P | Q) = \int_0^{+\infty} e^{-x} P(x) Q(x) dx$  est un produit scalaire sur  $\mathbb{R}[X]$ . Trouver une base orthogonale de  $\mathbb{R}_4[X]$ .

Une première possibilité est de travailler sur les polynômes. Malheureusement, MAPLE n'est pas fait pour traiter facilement le  $\lambda$ -calcul donc on ne peut pas créer une super-fonction de fonctions...

```
> ps := (P, Q) -> int(exp(-x)*P(x)*Q(x), x=0..infinity);
> B0 := [x -> x^0, x -> x^1, x -> x^2, x -> x^3, x -> x^4];
> P[1] := x -> B0[1](x);
> P[2] := x -> B0[2](x) - ps(P[1], B0[2])*P[1](x)/ps(P[1], P[1]);
> P[3] := x -> B0[3](x) - sum(ps(P[k], B0[3])*P[k](x)/ps(P[k], P[k]), k=1..2);
> P[4] := x -> B0[4](x) - sum(ps(P[k], B0[4])*P[k](x)/ps(P[k], P[k]), k=1..3);
> P[5] := x -> B0[5](x) - sum(ps(P[k], B0[5])*P[k](x)/ps(P[k], P[k]), k=1..4);
```

Une autre possibilité est de travailler sur les matrices et d'orthogonaliser la matrice du produit scalaire. Le produit scalaire des vecteurs s'obtient avec le point  $\cdot$  et  $\text{Column}(M,i)$  renvoie le  $i$ -ème vecteur colonne de  $M$ .

```
> M1 := matrix(5,5,(i,j) -> ps(B0[i],B0[j]));
```

**Exercice 4.** Équation différentielle Résoudre  $x(x^2-1)y' + (x^2+1)y + 2x^2 = 0$ . Donner les solutions sur  $\mathbb{R}$ .

Il faut juste manipuler les commandes basiques d'analyse :

```
> edo := x*(x^2-1)*diff(y(x),x)+y(x)*(x^2+1)+2*x^2;
> f:=x-> rhs(dsolve(edo,y(x)));
> f(x);
> limit(f(x),x=0);
> limit(f(x),x=1,left);
> limit(f(x),x=1,right);
> limit(f(x),x=-1,left);
> limit(f(x),x=-1,right);
> f1(x) = subs(_C1=2,f(x));
> f2(x) = subs(_C1=-2,f(x));
> limit(diff(f1(x),x),x=0);
> limit(diff(f2(x),x),x=0);
```

## C Un peu d'analyse

**Exercice 5.** Soit  $f : x \mapsto \frac{1}{x-\sin x} - \frac{1}{\frac{x^3}{6} - \frac{x^5}{120}}$ .

On veut obtenir un développement limité à l'ordre 5 au voisinage de 0.

Discuter de l'ordre des développements. Représentez graphiquement la fonction  $f$  et la fonction polynomiale associée au développement.

### Solution.

```
> f:=x->1/(x-sin(x))-1/(x^3/6-x^5/120);
> normal(f(x)); series(denom(%),x=0,7);
> series(f(x),x=0,12);
> p:=convert(%,polynom);
> g:=unapply(p,x);
> P1:=plot(f(t),t=0..4);
> P2:=plot(g(t),t=0..4,color=blue);
> plots[display](P1,P2);
```

**Exercice 6.** Étudier le comportement asymptotique de  $x \mapsto \frac{\sin(\frac{x+1}{x}) - \sin(\frac{x-1}{x})}{\sqrt{x^2+2} - \sqrt{x^2+1}}$ .

### Solution.

```
> f:=x->(sin((x+1)/x)-sin((x-1)/x))/(sqrt(x^2+2)-sqrt(x^2+1));
> asympt(f(x),x);
> evalf(%);
```

**Exercice 7.** a) Soit  $x \in ]-\frac{\pi}{2}; \frac{\pi}{2}[$ . On pose :

$$\begin{cases} f_1(x) = \frac{3 \sin(x)}{2 + \cos(x)}, & f_2(x) = \frac{1}{3} (8 \sin(\frac{x}{2}) - \sin(x)) \\ f_3(x) = \sqrt[3]{\sin^2(x) \tan(x)}, & f_4(x) = \frac{1}{3} (2 \sin(x) + \tan(x)) \end{cases}$$

Montrer l'existence d'un réel strictement positif  $\eta$  tel que l'on ait, pour tout  $x \in ]0; \eta[$ , la suite d'inégalités  $f_1(x) < f_2(x) < f_3(x) < f_4(x)$ .

b) On pose  $u(x) = 3(2 + \cos(x))(f_2(x) - f_1(x))$ . Montrer qu'il existe des réels  $a, b, c$  et  $d$  tels que l'on ait :

$$u(x) = a \sin(2x) + b \sin\left(\frac{3x}{2}\right) + c \sin(x) + d \sin\left(\frac{x}{2}\right)$$

Exprimer la dérivée  $u'(x)$  et vérifier qu'il existe  $P(y) \in \mathbb{R}_4[y]$  tel que l'on puisse écrire  $u'(x) = P(\cos(\frac{x}{2}))$ . En déduire sur  $]0; \frac{\pi}{2}[$  le signe de  $f_2(x) - f_1(x)$ .

### Solution.

```
> f[1]:=x->3*sin(x)/(2+cos(x)):f[2]:=x->1/3*(8*sin(x/2)-sin(x)):f[3]:=x->sin
(x)*(cos(x)^(-1/3)):f[4]:=x->1/3*(2*sin(x)+tan(x)):
> seq(f[k](x)=series(f[k](x),x=0),k=1..4);
> plot([seq(f[k](t),k=1..4)],t=0..1.5);
> u:=x->3*(2+cos(x))*(f[2](x)-f[1](x));
> v:=unapply(combine(u(x)),x);
> expand(diff(v(x),x));
> dv(t):=expand(subs(x=2*t,%));
> du(x):=subs(t=x/2,%);
> P(y):=subs(cos(x/2)=y,"");
> solve(P(y)>=0,y);
```

**Exercice 8.** Représenter graphiquement le système dynamique défini par une suite  $u_{n+1} = f(u_n)$  et la donnée de  $u_0$ . On fera figurer la courbe représentative de  $f$  ainsi que la première bissectrice et bien sûr l'escargot.

### Solution.

```
> escargot:=proc(f,u0,intervalle_x,intervalle_y,n)
local p,P,B,C,k,i;
p:=[[u0,u0],[u0,evalf(f(u0))]];
P:=plot([seq(seq(map(f@i,p[k]),k=1..2),i=1..n)],x=intervalle_x,y=
intervalle_y);
B:=plot(x,x=intervalle_x,y=intervalle_y,color=blue):
C:=plot(f(x),x=intervalle_x,y=intervalle_y,color=pink):
plots[display](P,C,B);
end;
```

### Exercice 9.

a) On considère la suite définie par  $u_0 \geq 0$  et  $u_{n+1} = \frac{1+u_n^2}{2}$ . Montrer que cette suite ne converge que si  $u_0 \leq 1$ .

b) Donner un équivalent puis un développement asymptotique à deux termes non nuls de la suite  $n \mapsto v_n = 1 - u_n$  pour  $u_0 = 0$ .

**Solution.**

```

> f:=x ->(1+x^2)*0.5:
> u:=proc(n,u0)
  option remember;
  if n=0 then u0
  else evalf(f(u(n-1,u0)))
  fi
end:
> u(100,0.):
> Es:=seq(escargot(f,0.1*k,0..3,0..3,5),k=0..15):
> plots[display](Es,insequence=true,view=[0..3,0..3]):
> solve(f(x)=x):

```

Pour le b), on utilise `rsolve` même si on n'obtient pas une expression explicite de  $v_n$  :

« *If rsolve is unable to compute a solution, it returns the unevaluated function invocation. This unevaluated rsolve invocation may be understood by other functions; for example, the asympt function is able to compute an asymptotic series expansion for the solution in some cases.* »

```

> v:=rsolve({a(n)=1-(1+(1-a(n-1))^2)/2,a(0)=0},a);
> asympt(v,n,3);

```

**Exercice 10.** (Centrale, 2010)

Trouver  $P \in \mathbb{R}[X]$  tel que la série de terme général  $u_n = (n^7 - 3n^6)^{1/7} - (P(n))^{1/3}$  converge le plus vite. Donner alors un équivalent du reste.

**Solution.** On cherche un polynôme de degré 3, puisqu'il faut annuler une expression équivalente à  $n$ .

```

u := (n^7-3*n^6)^(1/7);

# soit on trouve les coefficients à la main
asympt(u, n);
asympt(u - n, n);
asympt(u - (n^3 + c*n^2)^(1/3), n);
asympt(u - (n^3 - 9/7*n^2)^(1/3), n);
asympt(u - (n^3 - 9/7*n^2 + c*n)^(1/3), n);
asympt(u - (n^3 - 9/7*n^2 - 54/49*n)^(1/3), n);
asympt(u - (n^3 - 9/7*n^2 - 54/49*n + c)^(1/3), n);
asympt(u - (n^3 - 9/7*n^2 - 54/49*n -594/343)^(1/3), n);

# soit on écrit une boucle
P := n^3;
for i from 2 to 0 by -1 do
  unassign('c'):
  P := P + c*n^i;
  equiv := series(leadterm(u - P^(1/3)), n=infinity):
  c := solve(equiv, c):
  P := P;
end do;

# un équivalent du reste
series(leadterm(u - P^(1/3)), n=infinity);

```

Les calculs donnent

$$P(n) = n^3 - \frac{9}{7}n^2 - \frac{54}{49}n - \frac{594}{343}$$

On a alors

$$(n^7 - 3n^6)^{1/7} - (P(n))^{1/3} \Leftrightarrow -\frac{2673}{2401} n^{-3}$$

### Exercice 11.

- Soit  $y \in \mathbb{R}_+^*$ . Montrer qu'il existe un unique réel, noté  $f(y)$ , solution de l'équation  $xe^x = \frac{1}{y}$  d'inconnue  $x \in \mathbb{R}$ . Ceci définit  $f: \mathbb{R}_+^* \rightarrow \mathbb{R}$ .
- Tracer une approximation du graphe de  $f$  sur l'intervalle  $]0, 20]$ .
- Montrer que  $f$  est indéfiniment dérivable.
- Donner les deux premiers termes du développement limité de  $f$  en  $0^+$ .
- Donner les deux premiers termes du développement asymptotique de  $f$  en  $+\infty$ .

### Solution.

- a) Soit  $g: x \mapsto \frac{1}{xe^x}$ . On a  $g \in \mathcal{C}^\infty(\mathbb{R}_+^*, \mathbb{R}_+^*)$  et  $g' < 0$ , avec  $\lim_{x \rightarrow 0} g = +\infty$  et  $\lim_{x \rightarrow +\infty} g(x) = 0$  :

```
> diff(g(x), x);
> limit(g(x), x=0, right);
> limit(g(x), x=infinity);
```

Donc, pour tout  $y > 0$ , il existe un unique  $x > 0$  tel que  $y = g(x)$ .

- b) On introduit  $f$  sur Maple :

```
> f:=y->solve(x*exp(x)=1/y, x);
> plot(f(y), y=0..20);
```

- c) Comme  $g$  est un  $\mathcal{C}^1$ -difféomorphisme de  $\mathbb{R}_+^*$  dans lui-même, alors  $f$  est  $\mathcal{C}^\infty$ .

- d) On observe la limite de  $f$  avec Maple :

```
> limit(f(y), y=0);
```

Malheureusement, Maple ne nous aide pas à déterminer le développement asymptotique :

```
> series(f(y), y=0);
```

LambertW(1/y)

ce qui ne nous avance guère.

Or par définition de  $f$ , on a  $f(y)e^{f(y)} = \frac{1}{y}$ .

On en déduit que :

$$f(y) + \ln(f(y)) = -\ln(y)$$

Or  $\lim_{y \rightarrow 0} f(y) = +\infty$  donc  $\ln(f(y)) = o(f(y))$ . Il existe donc une fonction  $\varepsilon$  telle que :

$$f(y) = -\ln(y)(1 + \varepsilon(y)), \quad \lim_{y \rightarrow 0} \varepsilon(y) = 0$$

En reportant dans l'équation initiale on obtient :

$$-\ln(y) - \ln(y)\varepsilon(y) + \ln(-\ln(y)) + \ln(1 + \varepsilon(y)) = -\ln(y)$$

En remarquant que  $\ln(1 + \varepsilon(y)) \sim \varepsilon(y)$ , on aboutit à :



$$\varepsilon(y) \sim \frac{\ln(-\ln(y))}{\ln(y)}$$

puis :

$$f(y) = -\ln(y) - \ln(-\ln(y)) + o(\ln(-\ln(y)))$$

Ce n'est pas très rapide :

```
> h := y -> -ln(y) - ln(-ln(y)) :
> limit((f(y) - h(y)) / ln(-ln(y)), y = 0);
> plot([h(y), f(y)], y = 0..0.0000001);
```

e) On agit de même au voisinage de l'infini mais cette fois Maple nous aide :

```
> plot([h(y), f(y)], y = 0..0.0000001);
> asympt(f(y), y, 3);
```

### Exercice 12.

$$\mathbb{R} \rightarrow \mathbb{R}$$

On pose  $g: x \mapsto \begin{cases} \frac{\operatorname{ch}(x)}{\operatorname{sh}(x)} - \frac{1}{x} & \text{si } x \neq 0 \\ 0 & \text{sinon} \end{cases}$

- La fonction  $g$  est-elle continue? Soit  $\alpha \in \mathbb{R}^*$ . Soit  $f$  une fonction définie sur  $\mathbb{R}$ ,  $2\pi$ -périodique, telle que  $f(x) = \operatorname{ch}(\alpha x)$  pour  $x \in [-\pi, \pi]$ .
- Représenter  $f$  sur  $[-3\pi, 3\pi]$  pour  $\alpha = 2$ .
- Calculer la série de Fourier de  $f$  et étudier sa convergence.
- En déduire une expression de  $g$  comme somme de série de fonctions rationnelles.

### Solution.

a) Interrogeons Maple :

```
> g := x -> cosh(x) / sinh(x) - 1/x :
> limit(g(x), x = 0);
```

ce que confirme le DL :

```
> series(g(x) + 1/x, x = 0, 3);
```

b) On utilise Maple :

```
> h := x -> cosh(2*x) :
> f := unapply(piecewise(x < -Pi, h(x + 2*Pi), x < Pi, h(x), x < 3*Pi, h(x - 2*Pi)), x) :
> plot(f, -3*Pi..3*Pi);
```

c) Comme  $f$  est  $\mathcal{C}^1$  par morceaux et paire, la série de Fourier converge normalement. Pour les coefficients :

$$a_0(f) = \frac{1}{\pi} \int_0^{\pi} \operatorname{ch}(\alpha x) dx = \frac{\operatorname{sh}(\alpha\pi)}{\alpha\pi}$$

Pour les autres coefficients, deux intégrations par parties seront nécessaires.

On peut appeler Maple à la rescousse :

```

> f1(x) := cosh(a*x) : f2(x) := cos(n*x) :
> assume(n, integer) :
> J := simplify((2/Pi) * int(f1(x)*f2(x), x=0..Pi));
> combine(convert(J, trig));

```

C'est-à-dire

$$a_n(f) = \frac{2\alpha (-1)^n \sinh(\pi\alpha)}{\pi(n^2 + \alpha^2)}$$

Finalement :

$$\text{ch}(\alpha t) = \frac{\text{sh}(\alpha\pi)}{\alpha\pi} + \frac{2}{\pi} \alpha \text{sh}(\alpha\pi) \sum_{n=2}^{+\infty} \frac{(-1)^n \cos(nt)}{n^2 + \alpha^2}$$

d) On en déduit que :

$$\frac{\text{ch}(\alpha t)}{\text{sh}(\alpha\pi)} - \frac{1}{\alpha\pi} = 2\frac{\alpha}{\pi} \sum_{n=2}^{+\infty} \frac{(-1)^n \cos(nt)}{n^2 + \alpha^2}$$

Il suffit alors de prendre  $t = \pi$  et  $\alpha = \frac{x}{\pi}$  et on obtient :

$$g(x) = 2\frac{x}{\pi^2} \sum_{n=1}^{+\infty} \frac{(-1)^n (-1)^n}{n^2 + \frac{x^2}{\pi^2}} = 2x \sum_{n=1}^{+\infty} \frac{1}{\pi^2 n^2 + x^2}$$

### Exercice 13.

a) Résoudre l'équation différentielle

$$(E) : (1 + x^2)y' + xy - 2x = 0$$

b) Déterminer la solution particulière  $\varphi$  de (E) dont la courbe passe par  $A(0, 1)$ . et tracer cette courbe sur  $[-20, 20]$ .

c) Tracer la courbe précédente et la courbe obtenue au moyen de l'option `numeric`.

d) Donner un développement limité de  $\varphi$  à l'ordre 12 au voisinage de 0.

### Solution.

```

> with(plots):
> eq := (1+x^2)*diff(y(x), x) + x*y(x) - 2*x = 0:
> dsolve(eq, y(x)):
> dsolve({eq, y(0)=1}, y(x)):
> a := rhs("):
> A := plot(a, x=-20..20, color=black): display(A):
> b := dsolve({eq, y(0)=1}, y(x), numeric):
> B := odeplot(b, x=-20..20, color=blue, numpoints=500): display(A, B):
> Order:=13:
> dsolve({eq, y(0)=1}, y(x), series):

```

**Exercice 14.** On considère le système différentiel

$$(S) : \begin{cases} x' &= 3x + y + e^t \\ y' &= 2x + 2y + e^{2t} \end{cases}$$

- a) Résoudre ce système.  
 b) Tracer la courbe correspondant au problème de Cauchy (S) et  $x(0) = 1 - y(0) = 2$ .  
 c) Préciser l'allure de la courbe quand  $t$  tend vers  $-\infty$  et quand  $t$  tend vers  $+\infty$ .

**Solution.**

```
> sys:=diff(x(t),t)=3*x(t)+y(t)+exp(t),diff(y(t),t)=2*x(t)+2*y(t)+exp(2*t):
> dsolve({sys},{x(t),y(t)}):
> S:=dsolve({sys,x(0)=2,y(0)=-1},{x(t),y(t)}):
> X:=rhs(S[2]):Y:=rhs(S[1]):
> plot([X,Y,t=-10..10],view=[-2..10,-2..8],numpoints=500,scaling=constrained):
> limit(X,t=-infinity):limit(Y,t=-infinity):
> limit((2*X+2*Y+exp(2*t))/(3*X+Y+exp(t)),t=-infinity):
> limit(X,t=infinity):limit(Y,t=infinity):
> limit(Y/X,t=infinity):
> limit(Y-X,t=infinity):
```

**Exercice 15.** Déterminer les coefficients  $a, b, c, d$  et  $e$  pour que la fonction  $f(x) = \cos(x) + \frac{a+bx^2+cx^4}{1+dx^2+ex^4}$  ait un développement d'ordre le plus élevé possible au voisinage de 0. Donner un équivalent de  $f$ .

**Solution.**

```
> f:=cos(x)-(a+b*x^2+c*x^4)/(1+d*x^2+e*x^4);
> dev:=series(f,x=0,10);
> s:=convert(dev,polynomial);
> s:=solve({coeffs(s,x)},{a,b,c,d,e});assign(s):
> series(f,x=0,12);
> convert(%,polynomial);
```

**Exercice 16.** Soit  $f$  la fonction suivante :

$$f : (a, b) \mapsto \int_0^{+\infty} (x^2 + ax + b)^2 e^{-2x} dx$$

En quel(s) point(s)  $f$  atteint-elle son minimum ?

**Solution.**

```
> f:=(a,b)->int((x^2+a*x+b)^2*exp(-2*x),x=0..infinity):
> minimize(f(a,b)):
> #alternative
> mini:=solve({diff(f(a,b),a),diff(f(a,b),b)},{a,b}):assign(mini):
> f(a,b):
```

**Exercice 17.** (Centrale, PC 2010)

Soit, pour  $f \in \mathcal{C}^0([0, \pi], \mathbb{R})$  et  $n \in \mathbb{N}$ ,  $I_n(f) = \int_0^\pi f(t) |\sin nt| dt$

- a) Soit  $f: t \mapsto 1$ . Calculer  $I_n(f)$  pour  $n \in \llbracket 1, 10 \rrbracket$ . Prouver le résultat dans le cas général.

- b) Soit  $f: t \mapsto t$ . Calculer  $I_n(f)$  pour  $n \in \llbracket 1, 10 \rrbracket$ . Prouver le résultat dans le cas général.  
 c) Soit  $f: t \mapsto t^2$ . Calculer  $I_n(f)$  pour  $n \in \llbracket 1, 10 \rrbracket$ . Que peut-on conjecturer ?  
 d) Soit  $\varphi: t \mapsto |\sin t|$ . Montrer que  $\varphi$  est somme de sa série de Fourier.  
 En déduire, si  $f \in \mathcal{C}^0([0, \pi], \mathbb{R})$ , que  $(I_n(f))_{n \geq 0}$  tend vers une limite que l'on précisera.

**Solution.**

a) On a

$$\int_0^\pi |\sin nt| dt = \frac{1}{n} \int_0^{n\pi} |\sin u| du = \frac{1}{n} \sum_{k=0}^{n-1} \int_{k\pi}^{(k+1)\pi} |\sin u| du = \frac{1}{n} \sum_{k=0}^{n-1} \underbrace{\int_0^\pi |\sin u| du}_{=2} = 2$$

b) De même,

$$\int_0^\pi t |\sin nt| dt = \frac{1}{n^2} \int_0^{n\pi} u |\sin u| du = \frac{1}{n^2} \sum_{k=0}^{n-1} \int_{k\pi}^{(k+1)\pi} u |\sin u| du = \frac{1}{n^2} \sum_{k=0}^{n-1} \underbrace{\int_0^\pi (u + k\pi) |\sin u| du}_{=(2k+1)\pi} = \pi$$

c) On conjecture que  $I_n(t \mapsto t^2) \rightarrow \frac{2}{3}\pi^2$

d) La fonction  $\varphi$  est  $\pi$ -périodique, continue, de classe  $\mathcal{C}^1$  par morceaux ; donc d'après le théorème de convergence normale, la série de Fourier de  $\varphi$  converge normalement sur  $\mathbb{R}$  vers la fonction  $\varphi$ . Le calcul de ses coefficients donne :

$$\forall n \geq 1, \quad b_n(\varphi) = 0$$

$$a_0 = \frac{4}{\pi} \int_0^{\frac{\pi}{2}} \sin t dt = \frac{4}{\pi}$$

$$\begin{aligned} \forall n \geq 1, \quad a_n(\varphi) &= \frac{4}{\pi} \int_0^{\frac{\pi}{2}} \sin t \cos 2nt dt = \frac{4}{\pi} \int_0^{\frac{\pi}{2}} (\sin(2n+1)t - \sin(2n-1)t) dt \\ &= \frac{2}{\pi} \left[ -\frac{\cos(2n+1)t}{2n+1} + \frac{\cos(2n-1)t}{2n-1} \right]_0^{\frac{\pi}{2}} = -\frac{4}{\pi(4n^2-1)} \end{aligned}$$

D'où

$$\forall x \in \mathbb{R}, \quad \varphi(x) = \frac{2}{\pi} - \sum_{n=1}^{+\infty} \frac{4 \cos 2nx}{\pi(4n^2-1)}$$

En utilisant le théorème d'intégration terme à terme sur un segment, on montre facilement que

$$I_n(f) = \frac{2}{\pi} \int_0^\pi f(t) dt - \frac{4}{\pi} \sum_{p=1}^{\infty} \frac{1}{4p^2-1} \int_0^\pi f(t) \cos(2pnt) dt$$

On applique alors le théorème de convergence dominée pour les séries.

En effet, on a en notant  $u_p(n) = \frac{1}{4p^2-1} \int_0^\pi f(t) \cos(2pnt) dt$ , on a :

- \* pour tout  $p \geq 1$ ,  $u_p(n) \rightarrow 0$  par le lemme de Riemann Lebesgue ;
- \* pour tout  $p \geq 1$  et tout  $n \geq 0$ , on a

$$|u_p(n)| \leq \frac{1}{4p^2-1} \int_0^\pi |f(t)| dt$$

de plus, la série  $\sum_p \frac{1}{4p^2-1} \int_0^\pi |f(t)| dt$  converge absolument.

Donc d'après le théorème de convergence dominée pour les séries, on a

$$\forall f \in \mathcal{C}^1([0, \pi], \mathbb{R}), \quad I_n(f) \longrightarrow \frac{2}{\pi} \int_0^\pi f(t) dt$$

```
kernelopts(printbytes=false):

In := (f, n) -> int(f * abs(sin(n*t)), t=0..Pi);

# a)
f := 1;
for n to 10 do In(f, n); end do;
n := 'n'; assume(n, posint); In(f, n);

# b)
f := t;
for n to 10 do In(f, n); end do;
n := 'n'; assume(n, posint); In(f, n);

# c)
f := t^2;
for n to 10 do In(f, n); end do;

# avec une boucle...
for f in [1, t, t^2] do
    seq(In(f, n), n=1..10);
end do;

# autre variante
printlevel:=2;
for k from 0 to 2 do
    for n from 1 to 10 do
        printf("I%d(t^%d)=%a = %f\n", n, k, In(t^k, n), In(t^k, n));
    end do;
    printf("2/Pi*int(f,0..Pi) = %a = %g\n", int(f, t=0..Pi)*2/Pi, int(f, t=0..Pi)*2/Pi);
end do;

# test de la conjecture
f := t^2;
for n to 10 do
    In(f, 10*n), evalf(In(f, 10*n));
end do;
evalf(int(f, t=0..Pi)*2/Pi);

f := exp(t);
for n to 10 do
    evalf(In(f, 10*n));
end do;
evalf(int(f, t=0..Pi)*2/Pi);
```

**Exercice 18.** (Centrale, MP 2007)

Soit  $(x_n)_{n \geq 1}$  la suite définie par

$$x_1 > 0 \quad \text{et} \quad \forall n \in \mathbb{N}^*, \quad x_{n+1} = x_n + n/x_n$$

- a) Calculer avec Maple, les 10 premiers termes de la suite pour différentes valeurs de  $x_1$ . Commenter.  
 b) Minorer  $x_n$ . Si  $(y_n)_{n \geq 1}$  vérifie la même relation de récurrence, étudier  $x_n - y_n$ .  
 En déduire le comportement asymptotique de  $(x_n)$ .

**Solution.**

- a) On conjecture que  $x_n \rightarrow +\infty$  et que  $x_n \Leftrightarrow kn$ . En effet, si  $x_n \sim kn^\alpha$ , alors on doit avoir

$$\frac{x_{n+1}}{x_n} = \left(1 + \frac{1}{n}\right)^\alpha = 1 + \frac{\alpha}{n} + \mathcal{O}\left(\frac{1}{n^2}\right)$$

Donc il est intéressant d'examiner  $\lim_n n \left(\frac{x_{n+1}}{x_n} - 1\right)$

Ici, cette limite semble exister et valoir 1, donc  $x_n$  doit être équivalente à  $k \cdot n$ .

Dans le cas particulier où  $x_1 = 1$ , on remarque que  $x_n = n$  pour tout  $n$ , ce qui se vérifie par une récurrence facile.

- b) On montre par une récurrence facile que  $\forall n \in \mathbb{N}^*$ ,  $x_n > 0$ . Donc la suite est bien définie.  
 De plus, en appliquant l'inégalité  $2\sqrt{ab} \leq a + b$  avec  $a = x_n$  et  $b = n/x_n$ , on obtient

$$\forall n \in \mathbb{N}, \quad x_{n+1} \geq 2\sqrt{n} \quad \text{i.e.} \quad \forall n \in \mathbb{N}^*, \quad x_n \geq 2\sqrt{n-1}$$

Donc par comparaison,  $x_n \rightarrow +\infty$ .

Quitte à échanger le rôle de  $x_n$  et  $y_n$ , on peut supposer  $x_2 - y_2 > 0$ . On a alors

$$x_{n+1} - y_{n+1} = \left(1 - \frac{1}{x_n y_n}\right) \cdot (x_n - y_n)$$

Or

$$\forall n \geq 2, \quad 1 \geq 1 - \frac{1}{x_n y_n} \geq 1 - \frac{1}{4(n-1)} > 0$$

Donc la suite  $(x_n - y_n)_n$  décroît et strictement et est minorée par 0 ; donc elle converge, en vertu du théorème de limite monotone.

Donc  $x_n - y_n = o x_n$  i.e.  $x_n \Leftrightarrow y_n$

En prenant  $y_1 = 1$ , on en déduit

$$x_n \Leftrightarrow n$$

```

liste := proc(n, x1);
  x := evalf(x1);
  L := [x] ;
  for i to n do
    x := x + n/x;
    L := [op(L), x]
  end do;
  return L;
end proc;

for x1 in [.1, .47, .81, 1, 2.1, 10] do
  map(evalf, liste(10, x1));
end do;

# variante :
xn := proc(n, x1):
  x := evalf(x1);
  for i to n do
    x := x + n/x;
  end do;
end proc;

```

```

        end do;
    return x
end proc;

for x1 in [.1, .47, .81, 1, 2.1, 10] do
    for n in [100, 1000, 10000, 100000] do
        printf("%f ; ", n * (xn(n+1, x1) / xn(n, x1) - 1));
    end do;
    printf("\n");
end do;

```

**Exercice 19.** (Centrale, PSI 2010)

Soit  $(\mathcal{E})$ :  $y' - y = f$  où  $f \in C^0(\mathbb{R}, \mathbb{C})$  est  $2\pi$ -périodique.

- Résoudre  $(\mathcal{E})$  pour  $f = \cos$ ,  $f = \sin$  et  $f = \sin^2$ . Représenter les graphes des solutions avec comme condition initiale  $y(0) = 1$ .
- Montrer qu'il existe une unique solution bornée  $\varphi$ . Montrer que cette solution est périodique. Déterminer  $\varphi$  dans les trois cas précédents.

**Solution.**

- On trouve successivement :

$$y(x) = -1/2 \cos(x) + 1/2 \sin(x) + e^x \cdot C1$$

$$y(x) = -1/2 \cos(x) - 1/2 \sin(x) + e^x \cdot C1$$

$$y(x) = -1/2 + 1/10 \cos(2x) - 1/5 \sin(2x) + e^x \cdot C1$$

- La forme générale des solutions de l'équation est donnée par

$$y(x) = e^x \left( \lambda + \int_0^x e^{-t} f(t) dt \right) \quad \text{avec } \lambda \in \mathbb{C}$$

Si une telle solution est bornée, alors nécessairement l'expression entre parenthèse doit tendre vers une limite nulle en  $+\infty$ . Donc s'il existe une solution bornée sur  $\mathbb{R}$ , ce ne peut être que

$$\varphi(x) = e^x \int_{+\infty}^x e^{-t} f(t) dt$$

Montrons que  $\varphi$  est périodique, ce qui montrera que  $\varphi$  est bornée sur  $\mathbb{R}$ . On note  $T$  une période de  $f$ . Alors,

$$\forall x \in \mathbb{R}, \quad \varphi(x+T) = e^x e^T \int_{+\infty}^{x+T} e^{-t} f(t) dt = e^x e^T \int_{+\infty}^x e^{-u-T} \underbrace{f(u+T)}_{=f(u)} du = \varphi(x)$$

Donc  $\varphi$  est  $T$ -périodique et bornée.

Dans les trois cas particuliers envisagés par l'énoncé, on trouve successivement

$$\varphi(x) = 1/2 \cos(x) - 1/2 \sin(x)$$

$$\varphi(x) = 1/2 \cos(x) + 1/2 \sin(x)$$

$$\varphi(x) = 3/5 - 1/5 (\cos(x))^2 + 2/5 \sin(x) \cos(x)$$

```

plotsetup(maplet);

eq := diff(y(x), x) - y(x) = f;

for f in [cos(x), sin(x), sin(x)^2] do
  dsolve(eq, y(x));
  sol := rhs(dsolve({eq, y(0)=1}, y(x)));
  plot(sol, x=-10..4, -1..10);
end do;

# Variante :
for f in [cos(x), sin(x), sin(x)^2] do
  DEtools[DEplot](eq, y(x), x=-10..4, [[y(0)=1]], y=-1..10);
end do;

# Tracé des solutions bornées :
for f in [cos(t), sin(t), sin(t)^2] do
  phi := simplify(-exp(x) * int(exp(-t)*f, t=x..infinity));
  plot(phi, x=-10..10, -2..2);
end do;

```

## D Un peu d'algèbre

**Exercice 20.** (Centrale, 2010) *Polynômes de Tchébychev (1821–1894) de seconde espèce*

- Soit  $n \in \llbracket 0, 10 \rrbracket$ . Montrer en utilisant Maple qu'il existe  $P_n \in \mathbb{R}[X]$  tel que  $\forall t \in \mathbb{R}$ ,  $\sin(nt) = P_n(\cos t) \sin t$
- Soit  $n \in \mathbb{N}$ . Montrer qu'il existe un unique  $P_n \in \mathbb{R}[X]$  tel que  $\forall t \in \mathbb{R}$ ,  $\sin(nt) = P_n(\cos t) \sin t$
- Montrer que  $\forall n \in \mathbb{N}$ ,  $P_{n+2} = 2X P_{n+1} - P_n$
- Écrire une procédure permettant de calculer  $P_n$ .
- Montrer que l'application  $(P, Q) \mapsto \int_{-1}^1 P(t)Q(t)\sqrt{1-t^2} dt$  définit un produit scalaire sur  $\mathbb{R}[X]$ .
- Montrer que la famille  $(P_n)_{n \in \mathbb{N}}$  est une famille orthogonale de  $\mathbb{R}[X]$ .

### Solution.

a) cf. ci-dessous.

b) On utilise la formule de Moivre :

$$\begin{aligned}
 \forall t \in \mathbb{R}, \quad \sin(nt) &= \Im(e^{int}) = \Im((\cos t + i \sin t)^n) \\
 &= \sin t \cdot \sum_{k=0}^{\lfloor \frac{n-1}{2} \rfloor} (-1)^k \binom{n}{2k+1} \cos^{n-2k-1} t (1 - \cos^2 t)^k \\
 &\rightarrow P = \sum_{k=0}^{\lfloor \frac{n-1}{2} \rfloor} (-1)^k \binom{n}{2k+1} X^{n-2k-1} (1 - X^2)^k
 \end{aligned}$$

D'où l'existence.

Si deux polynômes conviennent, alors ils coïncident sur le segment  $] -1; 1[$ , donc ils sont égaux.



c) On part de la formule de trigonométrie :

$$\begin{aligned}\forall t \in \mathbb{R}, \quad \sin(n+2)t &= 2\sin(n+1)t \cos t - \sin nt \\ (\sin t)P_{n+2}(\cos t) &= 2(\sin t)P_{n+1}(\cos t) \cos t - (\sin t)P_n(\cos t)\end{aligned}$$

$$\text{Donc } \forall t \in ]0; \pi[, \quad P_{n+2}(\cos t) = 2(\cos t)P_{n+1}(\cos t) - P_n(\cos t)$$

$$\text{Donc } \forall x \in ]-1; 1[, \quad P_{n+2}(x) = 2xP_{n+1}(x) - P_n(x)$$

$$\text{Donc } P_{n+2} = 2XP_{n+1} - P_n$$

d) La forme donnée dans l'énoncé est bien définie et on montre facilement que cette forme est bilinéaire symétrique et définie positive (détails laissés aux soins du lecteur...).

e) Soit  $m, n \in \mathbb{N}$ . Alors

$$\begin{aligned}\langle P_n | P_m \rangle &= \int_{]-1; 1[} P_n(t)P_m(t)\sqrt{1-t^2} dt \\ &= \int_{]0; \pi[} P_n(\cos \varphi)P_m(\cos \varphi) |\sin \varphi| - \sin \varphi d\varphi \\ &= \int_{]0; \pi[} \sin n\varphi \cdot \sin m\varphi d\varphi \\ &= \frac{1}{2} \int_{]0; \pi[} (\sin(n-m)\varphi - \sin(n+m)\varphi) d\varphi \\ &= \frac{\pi}{2} \cdot \delta_{n,m}\end{aligned}$$

Donc la famille  $(P_n)_{n \in \mathbb{Z}}$  est orthogonale.

De plus,  $\|P_n\| = \sqrt{\frac{\pi}{2}}$ . Donc la famille  $\left(\sqrt{\frac{2}{\pi}} \cdot P_n\right)_{n \in \mathbb{Z}}$  est orthonormale.

```
# a)
for n from 0 to 10 do
  Q := expand(sin(n * t) / sin(t));
  P[n] := subs(cos(t)=x, Q);
end do;

# c)
tchebychev := proc(n)
  option remember;
  if n=0 then
    return 0
  elif n=1 then
    return 1
  else
    return expand(2*x*tchebychev(n-1) - tchebychev(n-2));
  end if;
end proc;

for n from 0 to 10 do
  evalb(tchebychev(n) = P[n]);
end do;

tchebychev(35);
```

**Exercice 21.** (Centrale, 2010)

On munit  $\mathbb{R}_n[X]$  du produit scalaire défini par  $\forall (P, Q) \in \mathbb{R}_n[X]^2, \langle P|Q \rangle = \int_{-1}^1 P(t)Q(t) dt$ .  
Soit  $L: P \in \mathbb{R}_n[X] \mapsto (aX^2 + bX - 1)P'' + (cX + d)P'$  avec  $(a, b, c, d) \in \mathbb{R}^4$ .

- Déterminer les réels  $a, b, c$  et  $d$  pour que  $L$  soit autoadjoint. On suppose dans la suite qu'ils sont choisis de cette façon.
- Lorsque  $n = 4$ , trouver les éléments propres de  $L$ .
- Montrer l'existence de  $\max \left\{ \frac{\langle P|L(P) \rangle}{\langle P|P \rangle} \mid P \in \mathbb{R}_n[X] \setminus \{0\} \right\}$   
Conjecturer sa valeur pour  $n \leq 10$ . Démontrer la conjecture.

### Solution.

- a) Le résultat fourni par Maple est  $\{d = 0, b = 0, a = 1, c = 2\}$  ce qui correspond à l'opérateur

$$L: P \in \mathbb{R}_n[X] \mapsto (X^2 - 1)P'' + 2XP' = ((X^2 - 1)P)'$$

Réciproquement, on vérifie que le quadruplet trouvé convient bien, puisque par intégration par parties, on a

$$\langle P|Q \rangle = \int_{-1}^1 (x^2 - 1)P(x)Q(x) dx$$

qui est bien symétrique en  $P$  et  $Q$

- b) On trouve 5 valeurs propres distinctes :

valeur propre	vecteur propre
6	$X^2 - \frac{1}{3}$
20	$X^4 - \frac{6}{7}X^2 + \frac{3}{35}$
12	$X^3 - \frac{3}{5}X$
0	1
2	$X$

- c) L'endomorphisme  $L$  est symétrique, donc il est orthogonalement diagonalisable.

Soit  $\mathcal{B} = (Q_0, \dots, Q_n)$  une base orthonormée de  $\mathbb{R}_n[X]$  constituée de vecteurs propres de  $L$ , associés aux valeurs propres  $\lambda_0 \leq \dots \leq \lambda_n$  (rangés dans l'ordre croissant).

Soit  $P \in \mathbb{R}_n[X]$  et  $\alpha_0, \dots, \alpha_n$  ses coordonnées dans  $\mathcal{B}$ ; on a alors

$$\frac{\langle P|L(P) \rangle}{\langle P|P \rangle} = \frac{\sum_{k=0}^n \lambda_k \alpha_k^2}{\sum_{k=0}^n \alpha_k^2} \leq \lambda_n \quad \text{avec égalité si } P = Q_n$$

Donc  $\max \left\{ \frac{\langle P|L(P) \rangle}{\langle P|P \rangle} \mid P \in \mathbb{R}_n[X] \setminus \{0\} \right\} = \max(L)$

```
# a)
ps := (P,Q) -> int(P * Q, x=-1..1):
L := P -> (a*x^2+b*x-1) * diff(P,x,x) + (c*x+d) * diff(P,x):
M := matrix(4, 4, (i, j) -> ps(x^(i-1), L(x^(j-1)))):
sol := solve({seq(seq(M[i, j] = M[j, i], i=1..j-1), j=1..4)});
assign(sol);
L;

# b)
with(LinearAlgebra):
```

```

f := (i, j) -> coeff(L(x^(j-1)), x, i-1);
ML := Matrix(5, f);
(vp, V) := Eigenvectors(ML);

# c)
conjecture := proc(n);
  ML := Matrix(n+1, f);
  (vp, V) := Eigenvectors(ML);
  maxi := max(seq(vp[i], i=1..n+1));
  for i to n+1 do
    Q[i] := add(V[j, i] * x^(j-1), j=1..n+1);
  end do;
  P := add(alpha[k] * Q[k], k=1..n+1);
  maxi2 := maximize(ps(P, L(P)) / ps(P, P), \
    [seq(alpha[k]=-infinity..infinity, k=1..n+1)]);
  return maxi, maxi2
end proc;

for n to 10 do
  conjecture(n);
end do;

```

**Exercice 22.** (Centrale, 2010)

Si  $n \geq 2$ , on note  $(e_1, \dots, e_n)$  la base canonique de  $\mathbb{R}^n$ .

Soit  $f_n \in \mathcal{L}(\mathbb{R}^n)$  tel que  $f_n(e_i) = e_{i+1}$  si  $i \in \llbracket 1, n-1 \rrbracket$  et  $f_n(e_n) = e_1$ .

- a) Écrire une procédure permettant d'obtenir la matrice  $A_n$  de  $f_n$  dans la base  $e$ .
- b) Dans cette question,  $n = 6$ .
  - i) Calculer le polynôme caractéristique de  $A_6$ . Factoriser ce polynôme sur  $\mathbb{R}[X]$ .  
On écrit  $\chi = \prod_{1 \leq k \leq p} R_k$  où les  $R_k$  sont irréductibles dans  $\mathbb{R}[X]$ .
  - ii) Déterminer  $\ker R_k(f_6)$  pour  $k \in \llbracket 1, p \rrbracket$ . Montrer que  $\mathbb{R}^6 = \ker R_1(f_6) \oplus \dots \oplus \ker R_p(f_6)$ .
  - iii) Donner la matrice de  $f_6$  dans une base adaptée à cette décomposition.
- c) Si  $n \geq 2$ , calculer le polynôme caractéristique  $\chi$  de  $A_n$ . Le factoriser sur  $\mathbb{R}[X]$ .

**Solution.**

$$\text{Mat}_e(f_6) = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad \text{Mat}_{e'}(f_6) = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

La matrice  $A_n$  se déduit de  $I_n$  par une permutation circulaire des colonnes vers la gauche. Donc il est clair que  $X^n - 1$  est un polynôme annulateur de  $f_n$ .

Pour montrer que  $\chi = X^n - 1$ , il suffit de développer le déterminant  $\det(xI_n - A_n)$  par rapport à la première ligne.

Enfin, on a

$$\chi = X^n - 1 = \begin{cases} (X-1)(X+1) \prod_{k=1}^{p-1} \left( X^2 - 2X \cos \frac{2k\pi}{2p} + 1 \right) & \text{si } n = 2p \\ (X-1) \prod_{k=1}^p \left( X^2 - 2X \cos \frac{2k\pi}{2p+1} + 1 \right) & \text{si } n = 2p+1 \end{cases}$$

```
with(LinearAlgebra);
# a)
A := proc(n);
  M := Matrix(n);
  for i to n-1 do
    M[i+1, i] := 1;
  end do;
  M[1, n] := 1;
  return M
end proc;

M := A(6);
I6 := IdentityMatrix(6);
latex(M);

# b)
# i)
chi := CharacteristicPolynomial(M, x);
chi := factor(chi);
R := op(chi);

# ii)
# Détermination des sev Ker(Rk(f))
F1 := NullSpace(M - I6);
F2 := NullSpace(M + I6);
F3 := NullSpace(M^2 + M + I6);
F4 := NullSpace(M^2 - M + I6);

P := <F1[1]|F2[1]|F3[1]|F3[2]|F4[1]|F4[2]>;
# Trois manières de vérifier que P est inversible :
Determinant(P);
NullSpace(P);
ColumnSpace(P);

# iii)
# Matrice de f_6 dans un base adaptée :
P^(-1).M.P;

# c)
for n from 2 to 10 do
  CharacteristicPolynomial(A(n), x), \
  factor(CharacteristicPolynomial(A(n), x));
end do;
```

**Exercice 23.** (Centrale, 2010)

- a) Démontrer que, si deux endomorphismes  $u$  et  $v$  d'un espace vectoriel  $E$  commutent, alors, les sous-espaces propres de  $u$  et l'image de  $u$  sont stables par  $v$ .

Dans les deux cas suivants :

$$A = \begin{pmatrix} 20 & 12 & -4 & 12 \\ -4 & -3 & 9 & -5 \\ -4 & 1 & 5 & -5 \\ -8 & -10 & 6 & -2 \end{pmatrix} \quad \text{et} \quad A = \begin{pmatrix} -12 & -16 & -8 & -4 \\ 4 & 13 & 1 & -1 \\ 4 & 5 & 9 & -1 \\ 8 & 10 & 2 & 6 \end{pmatrix}$$

- b) Préciser les matrices qui commutent avec  $A$  (structure, dimension, base éventuelle).  
 c) Etudier dans  $\mathcal{M}_4(\mathbb{R})$ , puis dans  $\mathcal{M}_4(\mathbb{C})$ , l'équation

$$X^2 = A$$

(nombre de solutions, un exemple de solution quand il y en a, somme et produit des solutions quand elles sont en nombre fini).

### Solution.

```
with(LinearAlgebra):
# a)
A := <<20|12|-4|12>, <-4|-3|9|-5>, <-4|1|5|-5>, <-8|-10|6|-2>>;
(vpA, P) := Eigenvectors(A);
P^(-1).A.P;

B := <<-12|-16|-8|-4>, <4|13|1|-1>, <4|5|9|-1>, <8|10|2|6>>;
(vpB, Q) := Eigenvectors(B);
Q^(-1).B.Q;

# c)
Determinant(A);
Determinant(B);
```

a) C'est du cours.

b) On charge le package permettant les manipulations linéaires puis on définit la matrice

On observe que dans le premier cas,  $A$  est diagonalisable et semblable à  $\text{diag}(8, 4, 12, -4)$ .

De même dans le deuxième cas,  $A$  est diagonalisable et semblable à  $\text{diag}(-4, 8, 8, 4)$ .

Puisque dans les deux cas la matrice  $A$  est diagonalisable, les matrices commutant avec  $A$  sont celles laissant stables les sous-espaces propres de  $A$ .

- \* Dans le premier cas, les matrices commutant avec  $A$  sont de la forme  $PDP^{-1}$  avec  $D$  matrice diagonale et  $P$  la matrice de passage dans une base de diagonalisation de  $A$ , obtenue précédemment avec Maple. L'ensemble de ces matrices est une sous-algèbre de dimension 4 de  $\mathcal{M}_n(\mathbb{K})[4]\mathbb{K}$ .
- \* Dans le deuxième cas, les matrices commutant avec  $A$  sont de la forme  $Q\Delta Q^{-1}$  avec  $\Delta$  diagonale par blocs de la forme suivante

$$\Delta = \begin{pmatrix} a & 0 & 0 & 0 \\ 0 & b & c & 0 \\ 0 & d & e & 0 \\ 0 & 0 & 0 & f \end{pmatrix}$$

et  $Q$  la matrice de passage dans une base de diagonalisation de  $A$ , obtenue précédemment avec Maple. L'ensemble de ces matrices est une sous-algèbre de dimension 6 de  $\mathcal{M}_n(\mathbb{K})[4]\mathbb{K}$ .

- c) \* Dans le premier cas, l'équation  $X^2 = A$  n'admet pas de solution dans  $\mathcal{M}_n(\mathbb{K})[4]\mathbb{R}$  puisque  $\det A < 0$ .

Étudions l'équation dans  $\mathcal{M}_n(\mathbb{K})[4]\mathbb{C}$ . Si  $X$  est solution, alors  $X$  commute avec  $A$ , donc  $X$  laisse stable chacun des sous-espaces propres de  $A$ , à savoir chacune des quatre droites propres de  $A$ . Donc la restriction de (l'endomorphisme canoniquement associé à)  $X$  à chacune de ces droites est une homothétie. Donc  $X$  est diagonalisable dans toute base de diagonalisation de  $A$ .

On obtient donc  $2^4 = 16$  solutions, qui sont de la forme

$$P \operatorname{diag}(\pm 2\sqrt{2}, \pm 2, \pm 2\sqrt{3}, \pm i\sqrt{2})$$

où  $P$  est la matrice de passage dans une base de diagonalisation de  $A$ , obtenue avec Maple.

- \* Même raisonnement : les solutions sont de la forme

$$Q \operatorname{diag}(\pm i2, S, \pm 2)Q^{-1}$$

où  $S \in \mathcal{M}_n(\mathbb{K})[2]\mathbb{C}$  vérifiant  $S^2 = I_2$  (i.e. une symétrie de  $\mathbb{C}^2$ ). Il y a donc une infinité de solutions.

**Exercice 24.** (Centrale) Soit  $G$  le sous-groupe de  $GL_2(\mathbb{R})$  engendré par les deux matrices  $S$  et  $T$  suivantes :

$$S = \begin{pmatrix} -1 & 0 \\ 0 & 1 \end{pmatrix}, T = \frac{1}{\sqrt{2}} \begin{pmatrix} -1 & 1 \\ 1 & 1 \end{pmatrix}$$

Rappelons que c'est le plus petit sous-groupe de  $GL_2(\mathbb{R})$  contenant  $S$  et  $T$ .

- a) Avec le logiciel de calcul formel, créer les matrices  $S$ ,  $T$ . Expliciter les éléments du groupe  $\langle R \rangle$  engendré par la matrice  $R = ST$  et préciser le cardinal de ce sous-groupe de  $G$ . Quelles sont les matrices  $SR$  et  $R^7S$  ?
- b) Montrer que tout élément de  $G$  est soit une puissance  $R^k$  de  $R$ , soit un produit  $R^kS$ . Préciser le cardinal  $n$  de  $G$ . Dresser la liste de tous les éléments de  $G$  et déterminer la nature géométrique des endomorphismes canoniquement associés dans l'espace euclidien  $\mathbb{R}^2$ . c) La transformation  $\Phi_S : g \mapsto S \cdot g$  définit une permutation de l'ensemble  $G$ . À l'aide du logiciel de calcul formel, dresser la séquence des éléments de  $G$  et de leurs images par  $\Phi_S$ . Quelle est la signature de la permutation de  $G$  (qu'on peut identifier à l'ensemble  $\{1, 2, \dots, n\}$ ) ainsi définie ?

### Solution.

- a) On définit les matrices  $S$  et  $T$  puis on calcule  $R$  :

```
S := Matrix(2, 2, [-1, 0, 0, 1]);
T := Matrix(2, 2, [-1, 1, 1, 1]) / sqrt(2);
R := S . T;
```

$$R = \frac{1}{\sqrt{2}} \begin{pmatrix} 1 & -1 \\ 1 & 1 \end{pmatrix}$$

La matrice  $R$  est la matrice d'une rotation d'angle  $\frac{\pi}{4}$  et donc vérifie  $R^8 = I_2$ . On en déduit :

$$\langle R \rangle = \{I_2, R, R^2, \dots, R^7\}$$

groupe cyclique de cardinal 8. On peut visualiser les éléments de  $\langle R \rangle$  en écrivant

$\text{seq}(R^k, k=0..7);$

On calcule  $SR$  et  $R^7S$  :

$S.R;$   
 $R^7.S;$

On constate que  $SR = R^7S = T$ .

b) Considérons  $H = \langle R \rangle \cup \langle R \rangle S$ .

$H$  est évidemment une partie de  $G$  contenant  $S$  et  $T$ . On établit aisément  $SR^\ell = R^{7\ell}S$  pour tout  $\ell \in \mathbb{Z}$ . On en déduit alors que  $H$  est stable par produit. On en déduit aussi que  $H$  est stable par passage à l'inverse car

$$(R^k S)^{-1} = S^{-1} R^{-k} = SR^{-k} = R^{-7k} S$$

Ainsi  $H$  est un sous-groupe inclus dans  $G$  contenant  $S$  et  $T$ . Or  $G$  est le plus petit sous-groupe contenant  $S$  et  $T$  donc  $G = H$ .

Il y a 8 éléments dans  $\langle R \rangle$ , l'application  $M \mapsto MS$  étant injective, il y a aussi 8 éléments dans  $\langle R \rangle S$ . Enfin les éléments  $\langle R \rangle$  sont distincts de ceux de  $\langle R \rangle S$  car de déterminants distincts.

On en déduit

$$G = \{I_2, R, R^2, \dots, R^7\} \cup \{S, RS, R^2S, \dots, R^7S\}$$

de cardinal  $n = 16$ .

La séquence de tous les éléments de  $G$  est

$\text{seq}(R^k, k=0..7), \text{seq}(R^k.S, k=0..7);$

Les endomorphismes canoniquement associés aux éléments  $R^k$  sont des rotations, plus précisément, les rotations d'angles  $\frac{k\pi}{4}$ .

Les endomorphismes canoniquement associés aux éléments  $R^k S$  sont des réflexions. L'axe de réflexion s'obtient en recherchant un vecteur propre associé à la valeur propre 1.

c) On obtient la séquence des images respectives de la séquence précédente donnant les éléments de  $G$  en écrivant

$\text{seq}(S.R^k, k=0..7), \text{seq}(S.R^k.S, k=0..7);$

La permutation de  $\{1, 2, \dots, 16\}$  correspondante est

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 & 11 & 12 & 13 & 14 & 15 & 16 \\ 9 & 16 & 15 & 14 & 13 & 12 & 11 & 10 & 1 & 8 & 7 & 6 & 5 & 4 & 3 & 2 \end{pmatrix}$$

Le nombre d'inversion de celle-ci est

$$8 + (14 + 13 + 12 + 11 + 10 + 9 + 8) + 0 + (6 + 5 + 4 + 3 + 2 + 1 + 0)$$

soit encore

$$(1 + 2 + \dots + 14) + 1 = 106$$

La permutation considérée est donc paire, i.e. de signature 1.

**Exercice 25.** Soient  $n, p$  et  $q$  trois naturels non nuls et deux applications linéaires  $u \in \mathcal{L}(\mathbb{R}^p, \mathbb{R}^q)$  et  $v \in \mathcal{L}(\mathbb{R}^p, \mathbb{R}^n)$ .

a) Démontrer qu'il existe une application linéaire  $w \in \mathcal{L}(\mathbb{R}^n, \mathbb{R}^q)$  telle que  $u = w \circ v$  si, et seulement

si, on a l'inclusion des noyaux

$$\ker(v) \subset \ker(u)$$

Dans ce cas, déterminer toutes les applications  $w$  qui conviennent.

b) Pour résoudre cette question, on utilisera un logiciel de calcul formel.

Soient  $A$  et  $B$  les matrices de  $\mathcal{M}_3(\mathbb{R})$  suivantes :

$$A = \begin{pmatrix} -2 & 1 & 1 \\ 8 & 1 & -5 \\ 4 & 3 & -3 \end{pmatrix} \text{ et } B = \begin{pmatrix} 1 & 2 & -1 \\ 2 & -1 & -1 \\ -5 & 0 & 3 \end{pmatrix}$$

Existe-t-il une matrice  $C \in \mathcal{M}_3(\mathbb{R})$  telle que  $A = CB$  ?

Déterminer toutes les matrices  $C$  solutions.

### Solution.

a) S'il existe  $w$  tel que  $u = w \circ v$  alors pour tout  $x \in \ker v$ ,  $u(x) = (w \circ v)(x) = w(v(x)) = w(0) = 0$  et donc  $x \in \ker u$ . Ainsi  $\ker v \subset \ker u$ . Inversement, supposons  $\ker v \subset \ker u$ . Soit  $H$  un supplémentaire de  $\ker v$  dans  $\mathbb{R}^p$  :  $H \oplus \ker v = \mathbb{R}^p$ .

On sait que la restriction  $v|_H$  de  $v$  au départ de  $H$  est un isomorphisme de  $H$  vers l'image de  $v$ . Soit  $K$  un supplémentaire de  $\text{Im } v$  dans  $\mathbb{R}^n$  :  $K \oplus \text{Im } v = \mathbb{R}^n$ .

Considérons ensuite  $w_0$  l'application linéaire définie par :

$$\forall x \in \text{Im } v, w_0(x) = u(v|_H^{-1}(x)) \text{ et } \forall x \in K, w_0(x) = 0$$

D'une part, pour tout  $x \in \ker v$ ,  $(w_0 \circ v)(x) = w_0(0) = 0 = u(x)$  car  $\ker v \subset \ker u$ .

D'autre part, pour tout  $x \in H$ ,  $(w_0 \circ v)(x) = (w_0 \circ v|_H)(x) = (u \circ v|_H^{-1} \circ v|_H)(x) = u(x)$ .

Puisque les applications linéaires  $w_0 \circ v$  et  $u$  coïncident sur les sous-espaces vectoriels supplémentaires  $\ker v$  et  $H$ , c'est deux applications sont égales et on peut donc écrire  $u = w_0 \circ v$ .

Soit  $w \in L(\mathbb{R}^n, \mathbb{R}^q)$ .  $w$  est solution de l'équation  $u = w \circ v$  si, et seulement si,  $w \circ v = u = w_0 \circ v$  soit encore  $(w - w_0) \circ v = 0$ . Or  $f \circ g = 0 \Leftrightarrow \text{Im } g \subset \ker f$  donc  $u = w \circ v \Leftrightarrow \text{Im } v \subset \ker(w - w_0)$ .

Par suite les solutions de l'équation  $u = w \circ v$  sont de la forme  $w_0 + f$  avec  $f \in L(\mathbb{R}^n, \mathbb{R}^q)$  vérifiant  $\text{Im } v \subset \ker f$ .

b) On définit les matrices étudiées :

$$A := \text{Matrix}(3, 3, [-2, 1, 1, 8, 1, -5, 4, 3, -3]);$$

$$B := \langle\langle 1, 2, -1 \rangle \langle 2, -1, -1 \rangle \langle -5, 0, 3 \rangle \rangle;$$

On détermine les noyaux de celles-ci

$$\text{Nullspace}(A);$$

$$\text{NullSpace}(B);$$

On observe que ces noyaux sont égaux à  $\text{Vect}((3, 1, 5))$  et donc  $\ker B \subset \ker A$ . Par l'étude qui précède transposée aux matrices, on peut affirmer que l'équation étudiée possède au moins une solution.

Pour construire une solution à cette équation, il suffit d'introduire une matrice  $B_0$  inversible coïncidant avec  $B$  sur un supplémentaire de  $\ker B$  et de considérer  $C_0 = AB_0^{-1}$ .



En prenant pour supplémentaire de  $\text{Ker } B$  l'espace  $\text{Vect}((1,0,0), (0,1,0))$ , la matrice  $B_0 =$

$$\begin{pmatrix} 1 & 2 & 0 \\ 2 & -1 & 0 \\ -5 & 0 & 1 \end{pmatrix}$$

convient.

Définissons-la et calculons  $C_0$  :

```
B0:=Matrix([[1,2,0],[2,-1,0],[-5,0,1]]);
C0:=A.B0^(-1);
```

On obtient  $C_0 = \begin{pmatrix} 1 & 1 & 1 \\ -3 & -7 & -5 \\ -1 & -5 & -3 \end{pmatrix}$ .

On peut vérifier l'exactitude de cette solution par le calcul

```
A-C0.B;
```

Les autres matrices solutions se déduisent de  $C_0$  par ajout d'une matrice  $E$  telle que  $\text{Im } B \subset \text{Ker } E$ .

On détermine l'image de  $B$

```
ColumnSpace(B);
```

On obtient  $\text{Vect}((1,0,-1), (0,1,-2))$ .

La condition  $\text{Im } B \subset \text{Ker } E$  donne les relations  $C_1 = C_3$  et  $C_2 = 2C_3$  sur les colonnes de  $E$  qui est donc une matrice de la forme

$$\begin{pmatrix} a & 2a & a \\ b & 2b & b \\ c & 2c & c \end{pmatrix}$$

## E Géométrie

**Exercice 26.** Dans un repère orthonormé, on considère les 4 points suivants :

$$A(1,2,3) \quad B(2,4,-5) \quad C(0,1,-6) \quad D(-1,0,7)$$

Trouver le centre et le rayon de la sphère passant par ces 4 points.

### Solution.

```
> a:=(1,2,3):b:=(2,4,-5):c:=(0,1,-6):d:=(-1,0,7):
> f:=(u,v,w)->(u-x)^2+(v-y)^2+(w-z)^2:
> s:=solve({f(a)=R,f(b)=R,f(c)=R,f(d)=R},{x,y,z,R});assign(s):
> centre=[x,y,z];
> rayon=sqrt(R);
```

**Exercice 27.** (Centrale, PC 2010) Soit  $(S)$  la surface d'équation  $\frac{x^2}{9} + y^2 - \frac{z^2}{4} = 1$

- Nature de  $(S)$ ? Donner un paramétrage de  $(S)$ . Tracer  $(S)$  avec Maple.
- Déterminer l'intersection de  $(S)$  avec le plan d'équation  $z = \alpha$ .

- c) Calculer le volume du solide défini par  $\left\{ M(x, y, z) \mid \frac{x^2}{9} + y^2 - \frac{z^2}{4} = 1 \text{ et } a \leq z \leq b \right\}$
- d) La surface  $(S)$  admet-elle des points singuliers ?
- e) Donner l'équation du plan tangent à  $(S)$  au point  $M_t = (3 \cos t, \sin t, 0)$ .

**Solution.**

- a)  $(S)$  est un hyperboloïde à une nappe; il admet le paramétrage

$$\begin{aligned} x(u, v) &= 3 \cdot \cos(u) \operatorname{ch}(v) \\ y(u, v) &= \sin(u) \operatorname{ch}(v) \quad \text{avec } (u, v) \in ]-\pi, \pi] \times \mathbb{R} \\ z(u, v) &= 4 \cdot \operatorname{sh}(v) \end{aligned}$$

- b) L'intersection de  $(S)$  avec le plan  $z = \alpha$  est une ellipse  $(\mathcal{E}_\alpha)$  d'équation

$$\begin{cases} \frac{x^2}{9} + y^2 = \frac{\alpha^2}{4} + 1 \\ z = \alpha \end{cases}$$

de demi-grand axe  $A = 3\sqrt{\frac{\alpha^2}{4} + 1}$ , de demi-petit axe  $B = \sqrt{\frac{\alpha^2}{4} + 1}$  et d'aire  $\mathcal{A}(\alpha) = \pi AB = 3\pi \left( \frac{\alpha^2}{4} + 1 \right)$  (déduite par affinité de l'aire d'un cercle).

- c) Pour déterminer le volume du solide, on effectue une sommation par piles :

$$V = \int_{z=a}^{z=b} \operatorname{aire}(\mathcal{E}_z) dz = \int_{z=a}^{z=b} 3\pi \left( \frac{\alpha^2}{4} + 1 \right) dz = \frac{\pi}{4} (b-a)(b^2 + ab + a^2 + 4)$$

- d) Le gradient de la fonction  $f: (x, y, z) \mapsto \frac{x^2}{9} + y^2 - \frac{z^2}{4} - 1$  ne s'annule qu'en  $(0, 0, 0)$  qui n'appartient pas à l'hyperboloïde; donc  $(S)$  est une surface régulière.

- e) Le plan tangent à  $(S)$  en  $M(3 \cos t, \sin t, 0)$  a pour vecteur normal  $\nabla f(3 \cos t, \sin t, 0) = 2 \cdot \begin{pmatrix} \frac{1}{3} \cos t \\ \sin t \\ 0 \end{pmatrix}$

Ce plan a donc pour équation

$$\frac{1}{3} \cos t (x - 3 \cos t) + \sin t (y - \sin t) = 0 \iff \frac{1}{3} \cos t \cdot x + \sin t \cdot y = 1$$

C'est le plan vertical tangent à l'ellipse de gorge.

```
plotsetup(maplet);
with(plots);

implicitplot3d(x^2/9+y^2-z^2/4=1, x=-5..5, y=-5..5, z=-5..5, grid
=[13,13,13]);

# on obtient un meilleur tracé en paramétriques
H1 := plot3d([3*cos(u)*cosh(v), sin(u)*cosh(v), 2*sinh(v)], u=0..2*Pi, v
=-2..2);

# Intersection avec z=2
P := plot3d([x,y,4], x=-10..10, y=-10..10):
display3d({H1, P}, scaling=constrained);
```

**Exercice 28.** (Centrale, PC 2010) Soit  $(\mathcal{E})$  la surface d'équation  $x^2 + y^2 + 4z^2 = 1$ .

- a) Représenter  $(\mathcal{E})$ .
- b) Soit  $R_t$  la rotation d'axe dirigé et orienté par  $\mathbf{u} = \begin{pmatrix} 4 \\ 3 \\ 0 \end{pmatrix}$  et d'angle  $t$ .  
Soit  $(x', y', z')$  l'image de  $(x, y, z)$  par  $R_t$ . Exprimer  $(x', y', z')$  en fonction de  $(x, y, z)$ .
- c) Déterminer une équation de  $(\mathcal{E}_t)$  l'image de  $(\mathcal{E})$  par  $R_t$ .

**Solution.**

a) La surface  $(\mathcal{E})$  est un ellipsoïde de révolution d'axe  $(Oz)$ .

b) Deux façons de procéder :

\* soit on utilise la formule de la rotation ; en notant  $\mathbf{a} = \frac{1}{5}\mathbf{u}$ , on a

$$\forall \mathbf{x} \in E, \quad R_t(\mathbf{x}) = (1 - \cos t)S_{\mathbf{x}\mathbf{a}} \cdot \mathbf{a} + (\cos t) \cdot \mathbf{x} + (\sin t) \cdot (\mathbf{a} \wedge \mathbf{x}) \quad (\text{car } \|\mathbf{a}\| = 1)$$

D'où

$$\text{Mat}_{\mathcal{C}}(R_t) = (1 - \cos t) \cdot \mathbf{a}^t \mathbf{a} + \cos t \cdot I_3 + \sin t \cdot \begin{pmatrix} 0 & 0 & 3/5 \\ 0 & 0 & -4/5 \\ -3/5 & 4/5 & 0 \end{pmatrix}$$

où  $\mathcal{C}$  désigne la base canonique.

\* soit on profite du fait qu'ici les coordonnées du vecteur  $\mathbf{u}$  soient particulièrement simples pour trouver une base  $\mathcal{B}$  orthonormale directe adaptée à la rotation, et on applique les formules de changement de base :

$$\text{Mat}_{\mathcal{B}}(R_t) = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos t & -\sin t \\ 0 & \sin t & \cos t \end{pmatrix} \rightarrow \text{Mat}_{\mathcal{C}}(R_t) = P \text{Mat}_{\mathcal{B}}(R_t)^t P \quad \text{où } P = \begin{pmatrix} 4/5 & -3/5 & 0 \\ 3/5 & 4/5 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

\* En utilisant l'égalité  $\begin{pmatrix} x \\ y \\ z \end{pmatrix} = {}^t P \begin{pmatrix} x' \\ y' \\ z' \end{pmatrix}$ , on en déduit une équation de  $(\mathcal{E}_t)$  :

$$-\frac{27}{25} (\cos(t))^2 x^2 + \frac{52}{25} x^2 + \frac{72}{25} (\cos(t))^2 xy - \frac{72}{25} xy + \frac{18}{5} \cos(t) \sin(t) xz \\ - \frac{48}{25} (\cos(t))^2 y^2 + \frac{73}{25} y^2 - \frac{24}{5} \cos(t) \sin(t) yz + 3 (\cos(t))^2 z^2 + z^2 - 1 = 0$$

```
plotsetup(maplet);

# a)
with(plots):

implicitplot3d(x^2+y^2+4*z^2=1, x=-1..1, y=-1..1, z=-4..4,\
grid=[13,13,13], scaling=constrained);

# on obtient un meilleur tracé en paramétriques
plot3d([cos(u)*cos(v), sin(u)*cos(v), 1/2*sin(v)], \
u=-Pi..Pi, v=-Pi..Pi, scaling=constrained);

# b)
with(LinearAlgebra):
A := <4/5, 3/5, 0>;
R := (1-cos(t))* A.Transpose(A) \
+ cos(t) * IdentityMatrix(3) \
+ sin(t) * <<0|0|3/5>, <0|0|-4/5>, <-3/5|4/5|0>>;

# variante :
```

```

R0 := <<1|0|0>, <0|cos(t)|-sin(t)>, <0|sin(t)|cos(t)>>;
P := <<4/5|-3/5|0>, <3/5|4/5|0>, <0|0|1>>;
Rt := P.R0.Transpose(P);
Equal(Rt,R);

# c)
M := <x, y, z>;
Mt := Transpose(Rt).M;
eq := x^2 + y^2 + 4*z^2 - 1;
eqt := simplify(subs({x=Mt[1], y=Mt[2], z=Mt[3]}, eq));

```

## F CENTRALE 2011

**Exercice 29.** (MP) Pour  $x > 0$ , on définit les suites de fonctions  $u(n, x)$  par  $u(0, x) = x$  et  $u(n+1, x) = u(n, x)^2 + u(n, x)$  et  $v(n, x) = \frac{\ln u(n, x)}{2^n}$ .

Calculer les 20 premiers termes des deux suites pour  $x = 2$ . Conjecture?

Montrer que la série de terme général  $v(n+1, x) - v(n, x)$  est convergente et en déduire l'existence de  $a(x)$  tel que  $v(n, x) - a(x) = o(1/2^n)$ .

Montrer que  $u(n, x) \underset{n \rightarrow +\infty}{\sim} \exp(2^n a(x))$ .

On admet que  $a(x) = \ln x + \sum_{n \geq 0} \frac{1}{2^{n+1}} \ln \left( 1 + \frac{1}{u(n, x)} \right)$ . Montrer que  $a$  est continue sur  $]0, +\infty[$ .

Tracer le graphe de  $x \rightarrow \ln x + \sum_{n=0}^{20} \frac{1}{2^{n+1}} \ln \left( 1 + \frac{1}{u(n, x)} \right)$  sur  $[0, 20]$ . Commenter.

Donner un développement asymptotique de  $a(x) - \ln x$  à deux termes à l'aide de MAPLE.

### Solution.

```

> u := proc(n, x)
> option remember;
> if n = 0 then evalf(x)
> else evalf(u(n-1, x)^2+u(n-1, x))
> fi;
> end;
> U := [seq(u(n, 2.), n=0..20)];
> V := seq(2^(-n)*ln(U[n+1]), n=0..20);
> Ux := [seq(u(n, x), n=0..20)];
> a := x -> ln(x) + sum(2^(-n-1)*ln(1+1./Ux[n+1]), n=0..20);
> plot(a(x), x=0..20);
> U := [seq(u(n, 20.), n=0..20)];
> asympt(sum(2^(-n-1)*ln(1+1./Ux[n+1]), n=0..5), x, 3);

```

MAPLE donne une idée du résultat à partir d'une approximation (on ne considère qu'une somme partielle, très partielle même).

```

> plot([a(x), ln(x)+0.5/x], x=0..100)

```

**Exercice 30.** (MP) Étudier la courbe  $C$  d'équation  $y^2 = x^3 - x$ .

Construire un polynôme  $F$  non nul (à deux indéterminées) que doit annuler un couple  $(a, b)$  de scalaires pour que les droites d'équation  $y = a(x+1)$  et  $y = bx$  aient en commun un point de  $C$ .

Étudier alors la courbe d'équation  $F(x, y) = 0$ . On mènera notamment l'étude locale en  $(0, 0)$ , celle des branches infinies et on donnera l'allure de la courbe.

**Solution.** On reconnaît bien sûr une courbe elliptique...

```
> implicitplot(y^2=x^3-x, x=-5..5, y=-5..5, numpoints=5000);
# compte-tenu de la symétrie par rapport à (0x) :
> Y := x->sqr(x^3-x);
> solve(D(Y)(x)=0, x);
# recherche du polynôme
> S := solve({v=a*(u+1), v=b*u}, {u, v});
> u := rhs(S[1]); v := rhs(S[2]);
> v^2-u^3+u; # on règle la cas a=b et a=0 ou b=0
> F := unapply((b-a)^3*(v^2-u^3+u)/(a*b), a, b);
> factor(F(x, y));
> implicitplot(F(x, y)=0, x=-3..3, y=-3..3, numpoints=20000);
> implicitplot(F(x, y)=0, x=-0.5..0.5, y=-0.5..0.5, numpoints=20000);
> factor(diff(F(x, y), y)); # on peut utiliser le théorème des fonctions
    implicites
```

On pose  $y = a_0 + a_1x + a_2x^2 + o(x^2)$ . On obtient  $a_0 = a_2 = 0$  et  $a_1 = 2$  ce qui correspond aux représentations graphiques de MAPLE.

Pour les branches infinies :

```
> S2:=solve(F(x, y)=0, y);
> asympt(S2[1], x);
> asympt(S2[2], x);
```

**Exercice 31.** (PC) Pour tout entier  $N \geq 2$ , on note  $S_N = \sum_{n=2}^N \frac{e^{-nx}}{n^2-1}$ . Tracer simultanément  $S_2, S_5, S_{10}$  et  $S_{20}$  : comment semble se comporter  $S_N$  en  $+\infty$ ? Pour quelles valeurs de  $x$  a-t-elle une limite quand  $N \rightarrow +\infty$ ? On notera  $f(x)$  cette limite.

Montrer que  $f$  est continue sur son ensemble de définition et calculer  $f(0)$ . Montrer que  $f$  est  $\mathcal{C}^2$  sur  $\mathbb{R}_+^*$ .

Calculer  $\int_0^{+\infty} f(x) dx$ .

**Solution.**

```
> S := (N, x) -> sum(exp(-n*x)/(n^2-1), n=2..N);
> plot([S(2, x), S(5, x), S(10, x), S(20, x)], x=-0.1..5, color=[red, blue, green, gray
    ]);
> S(N, 0);
> limit(S(N, 0), N=infinity);
```

**Exercice 32.** (PC) Montrer que l'équation  $e^x = x^n$  avec  $n > 2$  possède une unique solution  $x_n$  sur  $[0, n]$ .

Calculer les  $x_n$  pour  $n \in \{10p \mid p \in \llbracket 1, 10 \rrbracket\}$ . Que peut-on dire de la suite  $(x_n)$ ? Démonstration. Trouver un développement asymptotique de  $x_n$  à l'ordre 5.

Montrer que l'équation admet i,e unique solution  $y_n$  sur  $[n, +\infty[$ . Étudier la suite  $(y_n)$ .

**Solution.**

$$x_n = 1 + o(1) \text{ d'où } \ln x_n = \frac{x_n}{x} = \frac{1}{n} + o\left(\frac{1}{n}\right).$$

$$x_n = \exp\left(\frac{1}{n} + o\left(\frac{1}{n}\right)\right) = 1 + \frac{1}{n} + o\left(\frac{1}{n}\right).$$

$$\text{Ainsi } x_n - 1 = \frac{1}{n} + o\left(\frac{1}{n}\right).$$

On continue en divisant à chaque itération par  $n$ .

```
> seq(fsolve(exp(x)=x^(10*p),x),p=1..10);
> P:= [seq(plot(exp(x)-x^(n),x=1.05..2),n=1..50)]:
> plots[display](P,view=[1..1.5,-0.5..0.5]);
> f := x -> x/ln(x):
> normal(D(f)(x));
> solve(f(x)>0,x);
> solve(D(f)(x)>0,x);
> plot(f(x),x=1..2);
> seq(fsolve(f(x)=10*p,x,0..10*p),p=1..10);
> u := solve(f(x)=n,x);
> asympt(u,n,7);
> w := 1 + 1/n+3/(2*n^2);
> asympt(exp(w/n),n);
> asympt(exp(w/n),n);
> seq(fsolve(f(x)=10*p,x,10*p..infinity),p=1..10);
> P:= [seq(plot(f(x)-10*p,x=10*p..100*p),p=1..20)]:
> plots[display](P,view=[2..1500,-0.5..0.5]);
> z := n -> fsolve(f(x)=n,x,n..infinity);
> z(1000);
```

**Exercice 33.** (PC)  $\Phi(P) = \int_{-1}^1 \frac{P(x)}{\sqrt{1-x^2}} dx$  est-elle définie sur  $\mathbb{R}_{2n-1}[X]$  ?

Montrer que  $\Phi$  est une forme linéaire. Pour  $n = 4$ , calculer les images par  $\Phi$  des vecteurs de la base canonique.

Montrer que  $\Psi$  définie par  $\Psi(P) = \frac{\pi}{n} \sum_{k=1}^n P\left(\cos\left(\frac{(2k-1)\pi}{n}\right)\right)$  est une forme linéaire. Pour  $n = 4$ ,

calculer les images par  $\Psi$  des vecteurs de la base canonique. Conclure (?).

Montrer que la famille définie par :

$T_0 = 1, T_1 = X, T_{p+1}(X) = 2XT_p(X) - T_{p-1}(X)$  est une base de  $\mathbb{R}_{2n-1}[X]$ .

Montrer que  $T_p(\cos(t)) = \cos(pt)$  puis calculer  $\Phi(T_p)$  et  $\Psi(T_p)$ .

**Solution.**

T comme ? Voir un exercice précédent du concours 2010.

```
> Phi := P -> int(P(x)/sqrt(1-x^2),x=-1..1);
> B := [seq(Phi(x->x^i),i=0..2*4-1)];
> psi:= P -> (Pi/4)*sum(P(cos((2*k-1)*Pi/4)),k=1..4);
> C := [seq(simplify(psi(x->x^i)),i=0..2*4-1)];
> T:= proc(n,x)
  if n = 0 then 1
  elif n=1 then x
  else 2*x*T(n-1,x)-T(n-2,x)
  fi;
end:

> combine(T(5,cos(t)),trig);
> seq(Phi(x->T(i,x)),i=0..12);
> seq(psi(x->T(i,x)),i=0..12);
```

**Exercice 34.** (PSI) Montrer que  $I_n = \int_{-\infty}^{+\infty} \frac{1}{(3+t^2)^n} dt$  existe pour  $n \geq 1$  et calculer les trois premiers termes.

Trouver une formule de récurrence permettant de calculer  $I_n$  et calculer les vingt premiers termes.

Étudier la suite  $(I_n)$ . Quelle est la nature de la série de terme général  $I_n$  ?

**Exercice 35.** (PSI) Résoudre  $y'' = (1+x)y' + y$  avec  $y(0) = y'(0) = 1$ . Tracer des solutions pour différentes valeurs de  $y'(0)$ . On note  $I_n$  le nombre d'applications  $f$  de  $\{1, 2, \dots, n\}$  dans lui-même vérifiant  $f \circ f = \text{Id}$ . Déterminer une relation de récurrence entre  $I_{n+2}$ ,  $I_{n+1}$  et  $I_n$ . Calculer  $I_n$  pour  $n \in \{1, 2, \dots, 15\}$ .

Montrer que le rayon de convergence  $R$  de  $\sum_{n \geq 1} \frac{I_n}{n!} x^n$  est non nul.

Calculer la somme de cette série sur le disque ouvert de convergence.

**Exercice 36.** (PSI) Soit  $n \in \mathbb{N}^*$  et  $\Phi_n: P \in \mathbb{R}_n[X] \mapsto X^n \cdot P(X + 1/X)$

- Calculer  $\Phi_4(X^4 + 4X^3 - 2X^2 + X + 1)$  et son reste dans la division euclidienne par  $X^5$ .
- Si  $P \in \mathbb{R}_n[X]$  montrer que  $\Phi_n(P)$  est un polynôme de degré  $\leq 2n$ . On note  $\Psi_n(P)$  le reste de la division euclidienne de  $\Phi_n(P)$  par  $X^{n+1}$ . Montrer que  $\Psi_n$  est un endomorphisme de  $\mathbb{R}_n[X]$ .
- Pour  $k \in \llbracket 1, 4 \rrbracket$ , donner la matrice  $M_k$  de  $\Psi_k$  dans la base canonique de  $\mathbb{R}_k[X]$ ; déterminer ses éléments propres.

**Exercice 37.** (PSI)

- Soit  $A \in S_3^+(\mathbb{R})$ . Montrer qu'il existe  $M \in M_3(\mathbb{R})$  telle que  $A = {}^t M M$ . Trouver  $M$  pour  $A =$

$$\begin{pmatrix} 13 & -5 & 2 \\ -5 & 13 & 2 \\ 2 & 2 & 6 \end{pmatrix}$$

- Vérifier que  $f: (x, y, z) \mapsto \frac{22x^2 + 6xy + 12y^2 - 7xz + 6yz + 4z^2}{13x^2 - 10xy + 4xz + 13y^2 + 6z^2 + 4yz}$  est définie sur  $\mathbb{R}^3 \setminus \{(0, 0, 0)\}$ .

- Montrer qu'il existe  $C \in S_3(\mathbb{R})$  telle que  $\{f(x, y, z) \mid (x, y, z) \neq (0, 0, 0)\} = \left\{ \frac{{}^t X C X}{{}^t X X} \mid X \neq (0, 0, 0) \right\}$

- Montrer que  $f$  admet un minimum et un maximum que l'on précisera.

- Tracer les surfaces d'équation  $f(x, y, z) = a$  pour deux valeurs distinctes de  $a$ . Conjecture ?

**Exercice 38.** (PC) Si  $n \in \mathbb{N}$ , on note  $\sigma(n)$  le cardinal de  $\{(n_1, n_2) \in \mathbb{N}^2 \mid 2n_1 + 3n_2 = n\}$

- Calculer  $\sigma(n)$  pour  $n \in \llbracket 0, 20 \rrbracket$ .

- Soit  $f: x \mapsto \sum_{n=0}^{+\infty} \sigma(n)x^n$ . Montrer que  $f$  a un rayon  $> 0$ . Montrer que  $f$  est une fraction rationnelle.

**Exercice 39.** (PSI) Soit  $f: \mathbb{R} \rightarrow \mathbb{R}$  1-périodique et telle que  $\forall x \in [-1/2, 1/2] \quad f(x) = x^2$

- Exprimer  $f(x)$  pour  $x \in \mathbb{R}$  à l'aide de la partie entière. Tracer le graphe de  $f$  sur  $[-2, 2]$ .

- On pose  $S_n: x \mapsto \sum_{k=-n}^n 2^k f\left(\frac{x}{2^k}\right)$

Représenter sur un même graphe  $f$  et  $S_n$  pour  $n \in \llbracket 1, 5 \rrbracket$ . Que peut-on conjecturer ? Le prouver.

**Exercice 40.** (PSI) Soit  $G: x \mapsto \int_{\mathbb{R}} \frac{\cos(xt)}{1+t^2} dt$

- Justifier la définition de  $G$ . Calculer  $G(x)$  pour  $x > 0$  à l'aide de Maple.

- Représenter graphiquement  $G$  avec Maple. Montrer que  $G$  est continue et bornée.

- Montrer que  $\forall x \in \mathbb{R} \quad G(x) = \int_{\mathbb{R}} \frac{x \cos(t)}{x^2 + t^2} dt$ . En déduire que  $G$  est de classe  $\mathcal{C}^2$ .

d) Calculer  $\frac{\partial}{\partial x^2} \left( \frac{x}{x^2+t^2} \right) = \frac{\partial}{\partial t^2} \left( \frac{x}{x^2+t^2} \right)$ . En déduire que  $G'' - G = 0$ .

**Exercice 41.** (PSI) Soit  $\mathcal{S}$  la surface d'équation  $x^2 + y^2 + z^2 + xy + yz + zx - x + y - 3z = 0$

- a) Représenter  $\mathcal{S}$ . Trouver une équation réduite. Déterminer l'ensemble des cercles inclus dans  $\mathcal{S}$ .
- b) Soit  $u = (1, 1, 1)$ . Déterminer l'ensemble  $\Gamma_u$  des points de  $\mathcal{S}$  en lesquels le plan tangent contient  $u$ . Tracer  $\Gamma_u$ . Déterminer le cylindre constitué des droites passant par  $\Gamma_u$  et dirigées par  $u$ .